```
MW = MolarMass(N2)

"Inputs"
P_tpsi = 300

P_t = P_tpsi*convert(psi,psf)

V_tgal = 3

V_t = V_tgal*convert(gal,ft^3)

T_1 = 500

T_1max = 560

P_2 = 400

P_1 = 2000

Rme = R#*convert(Btu/lbmol-R,ft-lbf/lbmol-R)

R_g = Rme / MW

n = IsentropicExponent(N2,T=T_1)

"Isentropic expansion of nitrogen from P_1 psi to P_2 psi"
T_2 = T_1*(P_2/P_1)^((n-1)/n)

T_g = (T_2+T_1)/2

W_g = P_t*V_t/(R_g*T_g) "Net pressurant weight"

"H&H procedure for sizing pressurant tanks"
"Compute pressurant volume"
"gross gas requirement = net system pressurant requirement + residual gas in storage vessel"
P_1*convert(psi,psf)*V_L/(T_1*R_g) = W_g + P_2*convert(psi,psf)*V_L/(T_2*R_g)

"Compute gross pressurant weight - include 2% margin"
W_p = P_1*convert(psi,psf)*V_L/(R_g*T_1) * 1.02

"Compute required volume of the storage vessel"
V_u = R_g*T_1max*W_p/(P_1*convert(psi,psf))

V_ugal = V_u*convert(ft^3,gal)

"Sutton procedure for sizing pressurant tanks"
P_p = P_t
V_p = V_t
k = n
T_0 = T_1
P_g = P_2*convert(psi,psf)
P_0 = P_1*convert(psi,psf)

m_0 = P_p*V_p*k /(R_g*T_0*(1-P_g/P_0))

V_0 = m_0*R_g*T_0/P_0

V_0gal = V_0*convert(ft^3,gal)
```

SOLUTION

Unit Settings: Eng R psia mass deg

$k = 1.4$

$MW = 28.01$ [lbm/lbmol]

$m_0 = 1.1$ [lbm]

$n = 1.4$

$P_0 = 287998$ [psf]

$P_1 = 2000$ [psia]

$P_2 = 400$ [psia]

$P_g = 57600$ [psf]

$P_p = 43200$ [psf]

$P_t = 43200$ [psf]

$P_{tpsi} = 300$ [psi]

$Rme = 1545$ [ft-lbf/lbmol-R]

$R_g = 55.16$ [ft-lbf/lbm-R]

$T_0 = 500$ [R]

$T_1 = 500$ [R]

$T_{1max} = 560$ [R]

$T_2 = 315.6$ [R]

$T_g = 407.8$ [R]

$V_0 = 0.1053$ [ft^3]

$V_{0gal} = 0.7878$ [gal]

$V_L = 0.108$ [ft^3]

$V_p = 0.401$ [ft^3]

$V_t = 0.401$ [ft^3]

$V_{tgal} = 3$ [gal]

$V_u = 0.1234$ [ft^3]

$V_{ugal} = 0.9228$ [gal]

$W_g = 0.7702$ [lbm]

$W_p = 1.15$ [lbm]

No unit problems were detected.